

# Trabalho 4 - Parte 1

## Alunos

- Thiago Lamin de Souza.
- João Victor Assaoka Ribeiro.
- Lucas de Oliveira Bontempi.
- Miguel Rangel.
- Avelange Azevedo da Costa.

## a) Segue abaixo o problema de programação linear:

### Conjuntos e Índices

- $P$ : Conjunto de plantas, indexado por  $p$  e  $pp$ .
- $M$ : Conjunto de máquinas, indexado por  $m$ .
- $T$ : Conjunto de tipos de produto, indexado por  $t$ .
- $Tr$ : Conjunto de trimestres, indexado por  $tr$ .

### Conjuntos Derivados

- $M_p$ : Conjunto de máquinas localizadas na planta  $p \in P$ .
- $T_m$ : Conjunto de tipos de produto que a máquina  $m \in M$  pode produzir.
- $T_p$ : Conjunto de todos os tipos de produto que podem ser produzidos na planta  $p \in P$ .

### Parâmetros

- $d_{p,t,tr}$ : Demanda do produto  $t$  na planta  $p$  no trimestre  $tr$ .
- $c_{m,t}$ : Custo para produzir uma unidade do produto  $t$  na máquina  $m$ .
- $u_{m,t}$ : Dias de uso da máquina  $m$  para produzir uma unidade do produto  $t$ .
- $a_{m,tr}$ : Total de dias disponíveis da máquina  $m$  no trimestre  $tr$ .

- $ct_{p,pp}$ : Custo para transportar uma unidade de produto da planta  $p$  para a planta  $pp$ .
- $S_{p,tr}$ : Capacidade máxima de armazenamento na planta  $p$  no trimestre  $tr$ .
- $h_{p,t}$ : Custo de manuseio para armazenar uma unidade do produto  $t$  na planta  $p$ .

## Variáveis

- $x$ : Quantidade a ser produzida do tipo  $t$ , na planta  $p$ , na máquina  $m$  e no trimestre  $tr$ .
- $A$ : Quantidade a ser armazenada do tipo  $t$ , na planta  $p$ , no trimestre  $tr$ .
- $T$ : Quantidade a ser transferida da planta  $p$  para a planta  $pp$ , do tipo  $t$  e no trimestre  $tr$ .

## Função Objetivo

O objetivo é minimizar o Custo Total ( $Z$ ), que é composto pela soma dos custos de **\*\*produção\*\***, **\*\*manuseio\*\*** (armazenamento) e **\*\*transporte\*\***.

$$\begin{aligned}
Z = & \sum_{p,m,t,tr} (c_{m,t} \cdot x_{p,m,t,tr}) && \text{(Custo de Produção)} \\
& + \sum_{p,t,tr} (h_{p,t} \cdot A_{p,t,tr}) && \text{(Custo de Manuseio)} \\
& + \sum_{p,pp,t,tr} (ct_{p,pp} \cdot T_{p,pp,t,tr}) && \text{(Custo de Transporte)}
\end{aligned}$$

## Sujeito a

$$\begin{aligned}
\sum_{m \in M_p, t \in T_m} x_{p,m,t,tr} + \sum_{pp \neq p} T_{pp,p,t,tr} + A_{p,t,tr-1} &\geq d_{p,t,tr} + \sum_{pp \neq p} T_{p,pp,t,tr} + A_{p,t,tr} \\
\forall p \in P, \forall t \in T_p, \forall tr \in Tr &\quad \text{(Garante que a oferta atenda a demanda e defina o estoque; } A_{p,t,0} = 0)
\end{aligned}$$

$$\begin{aligned}
\sum_{t \in T_m} (u_{m,t} \cdot x_{p(m),m,t,tr}) &\leq a_{m,tr} \\
\forall m \in M, \forall tr \in Tr &\quad \text{(Limita o tempo de produção à disponibilidade da máquina)}
\end{aligned}$$

$$\begin{aligned}
\sum_{t \in T_p} A_{p,t,tr} &\leq S_{p,tr} \\
\forall p \in P, \forall tr \in Tr &\quad \text{(Assegura que o estoque não exceda a capacidade do armazém)}
\end{aligned}$$

$$x_{p,m,t,tr}, A_{p,t,tr}, T_{p,pp,t,tr} \geq 0 \quad \text{(Impede quantidades negativas de produção, estoque ou transporte)}$$

## b) Resolvendo via AMPL

### Código AMPL

```
# --- Conjuntos ---
set PLANTA;
set MAQUINA;
set TIPO;
set TRIMESTRE;

# --- Parâmetros ---
param planta_maquina {MAQUINA} integer;
param produz_tipo {MAQUINA, TIPO} binary;

# --- Conjuntos Derivados ---
set M_P {p in PLANTA} := {m in MAQUINA: planta_maquina[m] = p};
set TIPO_M {m in MAQUINA} := {t in TIPO: produz_tipo[m,t] = 1};
set TIPO_P {p in PLANTA} := union {m in M_P[p]} TIPO_M[m];

# --- Parâmetros de Custo e Capacidade ---
param demanda {PLANTA, TIPO, TRIMESTRE} >= 0;
param custo {m in MAQUINA, TIPO_M[m]} >= 0;
param dias_maquina {m in MAQUINA, TIPO_M[m]} >= 0;
param disponibilidade_maquina {MAQUINA, TRIMESTRE} >= 0;
param custo_transporte {PLANTA, PLANTA} >= 0, default 0;
param armazenamento {PLANTA, TRIMESTRE} >= 0;
param manuseio {PLANTA, TIPO} >= 0, default 0;

# --- Variáveis de Decisão ---
var x {p in PLANTA, m in M_P[p], t in TIPO_M[m], tr in TRIMESTRE} >= 0;
var A {p in PLANTA, t in TIPO_P[p], tr in TRIMESTRE} >= 0;
var T {p in PLANTA, pp in PLANTA diff {p}, t in TIPO_P[p], tr in TRIMESTRE: t in TIPO_P[pp]} >= 0;

# --- Função Objetivo ---
minimize gasto:
    sum {p in PLANTA, m in M_P[p], t in TIPO_M[m], tr in TRIMESTRE}
        (x[p, m, t, tr] * custo[m, t])
    +sum {p in PLANTA, t in TIPO_P[p], tr in TRIMESTRE}
        (A[p, t, tr] * manuseio[p, t])
    +sum {p in PLANTA, pp in PLANTA diff {p}, t in TIPO_P[p], tr in TRIMESTRE: t in TIPO_P[pp]}
        (T[p, pp, t, tr] * custo_transporte[p, pp]);

# --- Restrições ---
subject to atender_demanda {p in PLANTA, t in TIPO_P[p], tr in TRIMESTRE}:
    sum {m in M_P[p]: produz_tipo[m,t] = 1} x[p, m, t, tr]
    + sum {pp in PLANTA: pp != p and t in TIPO_P[p]} T[pp, p, t, tr]
    + (if tr > 1 then A[p, t, tr - 1] else 0)
    >= demanda[p, t, tr]
```

```

+ sum {pp in PLANTA: pp != p and t in TIPO_P[pp]} T[p, pp, t, tr]
+ A[p, t, tr];

subject to disponibilidade {m in MAQUINA, tr in TRIMESTRE}:
    sum {t in TIPO_M[m]} (x[planta_maquina[m], m, t, tr] * dias_maquina[m, t])
    <= disponibilidade_maquina[m, tr];

subject to capacidade_armazenamento {p in PLANTA, tr in TRIMESTRE}:
    sum {t in TIPO_P[p]} (A[p, t, tr]) <= armazenamento[p, tr];

```

## Resultado

```

gasto = 5061340

x [*,*,35,1] (tr)
:      1      2      3      4      :=
'1'      .      .      .      .
'70'      .      .      .      .
C4      199      .      .      .
D5      0      .      .      .
D6      0      .      .      .
J6      .      .      0      .
K4      .      .      32      .
U2      .      82      .      .

[*,*,35,2] (tr)
:      1      2      3      4      :=
'1'      .      .      .      .
'70'      .      .      .      .
C4      208      .      .      .
D5      0      .      .      .
D6      0      .      .      .
J6      .      .      0      .
K4      .      .      33      .
U2      .      84      .      .

[*,*,35,3] (tr)
:      1      2      3      4      :=
'1'      .      .      .      .
'70'      .      .      .      .
C4      206      .      .      .
D5      0      .      .      .
D6      0      .      .      .
J6      .      .      0      .
K4      .      .      34      .
U2      .      83      .      .

[*,*,35,4] (tr)
:      1      2      3      4      :=
'1'      .      .      .      .
'70'      .      .      .      .
C4      215      .      .      .
D5      0      .      .      .
D6      0      .      .      .
J6      .      .      0      .
K4      .      .      36      .
U2      .      84      .      .

[*,*,42,1] (tr)
:      1      2      3      4      5
'1'      .      .      .      134      .
'70'      .      .      .      .      .
C4      0      .      .      .      .
D5      284      .      .      .      .
D6      0      .      .      .      .
J6      .      .      0      .      .
K4      .      .      141      .      .
T      .      226      .      .      .
U2      .      0      .      .      .
V1      .      .      .      .      1168

[*,*,42,2] (tr)
:      1      2      3      4      5
'1'      .      .      .      116      .
'70'      .      .      .      .      .
C4      0      .      .      .      .
D5      278      .      .      .      .
D6      0      .      .      .      .
J6      .      .      0      .      .
K4      .      .      160      .      .
T      .      255      .      .      .
U2      .      0      .      .      .
V1      .      .      .      .      1138

[*,*,42,3] (tr)
:      1      2      3      4      5
'1'      .      .      .      126      .
'70'      .      .      .      .      .
C4      0      .      .      .      .
D5      305      .      .      .      .
D6      0      .      .      .      .
J6      .      .      0      .      .
K4      .      .      175      .      .
T      .      272      .      .      .
U2      .      0      .      .      .
V1      .      .      .      .      1204

[*,*,42,4] (tr)
:      1      2      3      4      5
'1'      .      .      .      130      .
'70'      .      .      .      .      .
C4      0      .      .      .      .
D5      322      .      .      .      .
D6      0      .      .      .      .
J6      .      .      0      .      .
K4      .      .      188      .      .
T      .      289      .      .      .
U2      .      0      .      .      .
V1      .      .      .      .      1206

T [*,*,35,1]
:      1      2      3      4
1      .      0      0      61
2      0      .      0      0
3      0      0      .      0
4      0      0      0      .

[*,*,35,2]
:      1      2      3      4
1      .      0      0      66
2      0      .      0      0
3      0      0      .      0
4      0      0      0      .

[*,*,35,3]
:      1      2      3      4
1      .      0      0      67
2      0      .      0      0
3      0      0      .      0
4      0      0      0      .

[*,*,35,4]
:      1      2      3      4
1      .      0      0      75
2      0      .      0      0
3      0      0      .      0
4      0      0      0      .

```

## Discussão

A resolução do modelo via AMPL fornece o plano ótimo de produção, armazenamento e logística para minimizar os custos totais ao longo dos trimestres. Tudo o que não foi mostrado possui valor zero, o que inclui a variável A, que representa o quanto deveria ser armazenado numa determinada planta, de um certo tipo e num certo trimestre. Ou seja, a solução ótima sugere que nada seja armazenado, possivelmente por que não deve existir um gargalo na demanda, de maneira que seria necessário produzir mais em um trimestre e estocar (ou custos diferentes por trimestre). Além disso, nota-se que em todos os trimestres ocorreu a transferência de todos os recipientes demandados da planta 1 para a planta 4, isso provavelmente deve-se ao fato de que o custo de produção em 4 é mais

alto que o custo de produzir em 1 adicionado ao custo de transportar, além da planta em questão ter capacidade ociosa disponível.